

Simuler et visualiser un emballement thermique

J'invite le lecteur à consulter le site pour des informations complémentaires.

Page d'accueil du site Internet :

[page d'accueil](#)

d'autres pdf, sur différents sujets :

[liste des PDE](#)

Un emballement thermique est un phénomène d'augmentation de température divergent, généralement destructeur pour le composant. Ce document montre que ce phénomène est simulable avec Pspice.

Le composant choisi est une diode de puissance. En ayant la température comme variable, on peut simuler, sur le plan thermique si la condition est validée, le déclenchement de l'emballement thermique.

1) Comportement d'une diode vis à vis de la température

1.a Equation simplifiée du courant dans une jonction PN polarisée

→ Rappelons la loi de Shockley, qui donne le courant I en fonction de la tension V :

$$I = I_S \left[\exp\left(\frac{V}{N V_T}\right) - 1 \right] \quad I_S \text{ (en A) et } N \text{ (sans dimension) sont propres à chaque diode.} \quad [1]$$

Où $V_T = kT/q$, tension thermodynamique (en V). On a : $k = 1,38 \times 10^{-23}$ $q = 1,6 \cdot 10^{-19}$.

La température de jonction intervient ici par le terme V_T .

Sur cette loi *seule*, à tension V donnée, quand la température T augmente, le courant I diminue. C'est la conséquence d'avoir le terme V_T au dénominateur de la loi en exponentielle.

→ Mais le terme I_S est également lié à la température par des équations de la physique du semi-conducteur. La documentation de Pspice indique l'équation de I_S en fonction de T :

$$I_S(T) = I_S \exp \left[\left(\frac{T}{T_{nom}} - 1 \right) \left(\frac{E_G}{N V_T} \right) \right] \left(\frac{T}{T_{nom}} \right)^{\frac{XII}{N}}. \quad [2]$$

Remarquons que les paramètres d'un composant ont été calculés (ou mesurés) à la température nominale T_{nom} . Par défaut $T_{nom} = 300$ K. On peut vérifier que $I_S(300) = I_S$.

→ E_G est l'énergie de band gap. Pour le silicium, $E_G = 1,11$ eV. Mais cette valeur est aussi à pondérer en fonction de la température.

$$\text{Pspice propose :} \quad E_G(T) = 1,16 - 702 \cdot 10^{-6} \frac{T^2}{T + 1108}. \quad [3]$$

Il faut donc exploiter 3 équations pour tenir compte de la température de jonction.

Dans un but de simplification, on va se limiter à une diode en direct et négliger les phénomènes inverses, qui sont identifiés par d'autres équations et d'autres paramètres. Dans ce cas, on a le courant diode qui peut se déterminer, pour une température donnée, par l'équation [3] qui permet d'avoir E_G , puis par [2] pour avoir I_S , et enfin, avec la tension V, d'exploiter la loi [1].

Au bilan, à tension directe V donnée, le courant augmente avec la température : c'est essentiellement dû à l'équation [2].


Remarque : ces équations forment un modèle, et comme tout modèle, il y a un domaine de validité. Ici, la gamme de température doit rester « industrielle », comme -65 °C à 200 °C. Ces équations n'ont plus de raison d'être si la température est très éloignée de cette dynamique.

La résistance d'accès.

Entre les broches de la diode et la jonction PN interne, il y a une (très faible) résistance d'accès, nommée R_s par Pspice.

Sa valeur est donnée par le fabricant et fait partie des paramètres de la diode concernée.

Il faut donc tenir compte de la chute de tension dans R_s pour l'équation de la tension V , ce qui modifie le calcul, surtout à fort courant. Par exemple, à $10\text{ m}\Omega$ et 10 A , cela forme une chute de tension de 100 mV ce qui réduit d'autant la tension intrinsèque V .




1.b Introduire la température comme variable

Pspice permet de simuler en considérant la température comme **paramètre** (température de tel ou tel composant, ou de tout le circuit), mais pas comme **variable** (température qui évolue dans le temps), et n'analyse pas l'échauffement créé par un composant sur un autre, ni même l'auto échauffement.

La parade est de créer un composant « diode maison » où on peut représenter la température de jonction par une source de tension. Pour être au plus simple : 1 V représente $1\text{ }^\circ\text{C}$. Cette source de tension peut être fixe, ou variable dans le temps.

Pour simuler cette diode commandée par la température, on peut exploiter un sous circuit à base de sources contrôlées, dont les lois de commandes sont [1], [2] et [3], qui génère un courant entre A et K et qui est commandé par la source de tension, image de la température T .

Bien que possible, on ne va pas tenir compte de l'évolution de R_s vis à vis de la température.



1.c La diode choisie


Pour illustrer le comportement d'une diode en fonction de sa température, on choisit une diode de puissance : MUR810 de ON semiconductor.

Elle est spécifiée de $-65\text{ }^\circ\text{C}$ à $175\text{ }^\circ\text{C}$.

Elle est donnée pour 8 A max à $150\text{ }^\circ\text{C}$ de température de boîtier.

Le boîtier est TO 220 AC.

Les résistances thermiques sont $R_{THJC} = 3\text{ }^\circ\text{C/W}$ et $R_{THJA} = 73\text{ }^\circ\text{C/W}$.



Le modèle de cette diode est dans la bibliothèque diode.lib. Ses paramètres sont :

$I_s=4.433p$ $R_s=5.91m$ $I_{kf}=92.2m$ $N=1$ $X_{ti}=3$ $E_g=1.11$ $C_{jo}=265.3p$
 $M=.4253$ $V_j=.75$ $F_c=.5$ $I_{sr}=49.76n$ $N_r=2$ $T_t=61.66n$.

2) Simulation en température du sous circuit modélisant la diode MUR810

Si on ne considère que les équations [1], [2] et [3], il nous faut prendre ces 5 paramètres :

$I_s=4.433p$ $R_s=5.91m$ $N=1$ $X_{ti}=3$ $E_g=1.11$

Mais une diode ainsi simplifiée aurait une caractéristique un peu éloignée de la diode réelle, qui intègre d'autres phénomènes, comme les phénomènes inverses.

D'où, après quelques runs d'essais non présentés ici, on propose un jeu de valeurs qui permet d'avoir une bonne coïncidence entre la caractéristique donnée par le constructeur et la simulation Pspice, et ce dans une dynamique [$1\text{ A} - 30\text{ A}$] : $I_s=5.5p$ $R_s=9.5m$ $N=1.095$ $X_{ti}=4$ $E_g=1.11$


2.a Comparaison sur une simulation statique, des caractéristiques I = f (VAK).

(netliste en annexe 1 : diode_temperature.cir).

On réalise un circuit formé d'une source de tension qui alimente un sous circuit Xdiode qui reprend les équations [1], [2] [3].



On choisit le mode différentiel pour connecter la source de tension représentant la température.

(Des résistances de pull down de 1 MΩ ont été placées sur com+ et com- pour éviter des potentiels flottants et un message d'erreur de Pspice).



Par un run .DC, on fait un balayage continu de la tension V(1), ici entre 0,2 V et 1,1 V.

On réalise 3 balayages : à T = 25 °C, 100 °C, 175 °C et on compare avec ce que donne ON SC :




Interprétation

- On n'observe que la dynamique [1 A - 30 A], avec les mêmes échelles semi log.
- Pour chaque température, les caractéristiques courant-tension de la diode MUR810 et du sous circuit I(Xdiode) sont quasiment similaires. On peut ainsi dire qu'on a « refabriqué » un modèle « maison » de diode, avec la température comme variable d'entrée (tension entre les électrodes com+ et com-).
- Il est très visible que le courant augmente avec la température, à V donné.

2.b Caractéristique P = f (V, T)

Une courbe intéressante que l'on peut également afficher est la puissance P fonction de la tension externe et de la température, ici en échelle linéaire :



Interprétation

La puissance est donnée, sous probe, par W(Xdiode). C'est, en fait, la même expression que le produit $V(1) \times I(Xdiode :1)$.

A tension V constante, le courant circulant dans la diode, et donc la puissance dissipée par la jonction, augmentent avec la température.

Un zoom montre

qu'à $V = 0,6 \text{ V}$, $P = 4,5267 \text{ mW}$ à $25 \text{ }^\circ\text{C}$ et $1,6265 \text{ W}$ à $175 \text{ }^\circ\text{C}$.


Trace name	600.000m	700.000m
X Values	600.000m	700.000m
W(Xdiode)	4.5267m	170.322m
W(Xdiode)	223.561m	2.1518
W(Xdiode)	1.6265	5.7888

2.c Commande de l'évolution de la température de jonction

Elaborons un nouveau schéma où la tension d'alimentation de la « diode maison » est de $0,6 \text{ V}$, constante.

(netliste en annexe 2 : diode_commandee.cir)


On choisit une source « EXP » de Pspice pour représenter une évolution de température.



Alimentée sous 0,6 V, le sous circuit de comportement similaire à la diode MUR810 est soumis à une température variable dans le temps.

On simule une évolution de température d'un milieu telle que la température de jonction part de $25 \text{ }^\circ\text{C}$ et monte à $175 \text{ }^\circ\text{C}$ en suivant une loi exponentielle. Après stabilisation, cette température revient à sa valeur de départ en loi exponentielle décroissante, avec la même constante de temps.

On choisit arbitrairement une constante de temps de 100 ms .



Température, courant, puissance en fonction du temps

Interprétation :

Pour mieux voir le départ des courbes, le démarrage de la montée en exponentielle commence à $t = 0,1 \text{ s}$. La constante de temps étant de $0,1 \text{ s}$, le régime permanent est atteint à 99 % après $0,5 \text{ s}$. On choisit alors de faire retomber la température à $t = 1 \text{ s}$.

Le courant dans le sous circuit évolue de la valeur initiale ($7,54 \text{ mA}$, pour $T = 25 \text{ }^\circ\text{C}$) et atteint sa valeur finale qui vaut $2,71 \text{ A}$ quand $T = 175 \text{ }^\circ\text{C}$.

La puissance dissipée part de $4,53 \text{ mW}$ (soit $0,6 \text{ V} \times 7,54 \text{ mA}$) à $1,626 \text{ W}$ (soit $0,6 \text{ V} \times 2,71 \text{ A}$).


X Values	50.000m	950.000m
W(Xdiode)	4.5267m	1.6257
I(Xdiode:1)	7.5445m	2.7094
V(com+)- V(com-)	25.000	174.969

On observe nettement une distorsion entre la montée en température de la jonction et l'évolution du courant. Cela s'explique aisément par la loi – non linéaire – entre courant diode et température jonction.

2.c Loi d'Ohm thermique et point de fonctionnement statique

Rappelons la loi d'Ohm thermique, issue de la relation de Fourier, en conduction, exploitant le flux de chaleur :

Le flux est homogène à des W. On peut donner une formulation simple à cette relation : $\Delta T = R_{TH} P$ ΔT en °C, R_{TH} en 100°/W, P en W



L'analogie avec la loi d'Ohm électrique est évidente. Mais la grandeur qui « circule » est ici la puissance.

Appliquée à notre diode, cela forme : $\Delta T = \text{Température de Jonction} - \text{Température Ambiante} : T_J - T_A$. R_{TH} est la résistance thermique entre la jonction et l'air ambiant. En absence de dissipateur, la diode MUR810 présente $R_{THJA} = 73 \text{ °C/W}$.


Dans le quadrant P, T_J , cette loi d'ohm thermique est :

$P = (T_J - T_A) / R_{TH}$: c'est une droite, de pente $1/R_{TH}$, passant par 0 pour $T_J = T_A$.

On dispose, de la simulation précédente, la courbe $P = f(T)$: il suffit de placer la température de jonction en abscisse, soit $V(\text{com+}) - V(\text{com-})$, et d'afficher la puissance, par $W(\text{Xdiode})$ pour obtenir cette courbe. On peut également faire un balayage DC de $V(\text{com+}) - V(\text{com-})$, ce qui reste statique.

Pour illustrer la notion de stabilité thermique, on superpose sur cette dernière courbe, donc sur le même quadrant, la loi d'Ohm thermique.

Supposons une température ambiante de 60 °C, à proximité immédiate de la diode.



**Puissance dissipée par la diode = f (température de jonction) à $V = \text{constant} = 0,6 \text{ V}$.
On ajoute une droite $P = (T_J - T_A)/R_{TH}$ avec $T_A = 60 \text{ °C}$ $R_{TH} = 73 \text{ °/W}$.**

Ces 2 équations présentent 2 points communs :

X Values	63.000	173.080
W(Xdiode)	42.986m	1.5730

- [1] Le point de contact à $T_J = 63 \text{ °C}$, $P = 43 \text{ mW}$ est un point de fonctionnement STABLE : si la température augmente, la puissance délivrée par la jonction augmente, mais la puissance dissipable grâce à la loi d'Ohm thermique augmente plus.

- [2] A contrario, l'autre point, à $T_J \approx 173 \text{ °C}$, $P \approx 1,57 \text{ W}$, est INSTABLE : si la température augmente, la puissance délivrée par la jonction augmente, mais la puissance dissipable grâce à la loi d'Ohm thermique augmente plus faiblement. L'environnement thermique ne peut pas évacuer cet excès de température, et la diode va monter en température indéfiniment, jusqu'à sa destruction.

Dès lors, on comprend la notion de $R_{THlimite}$: si R_{THJA} augmente, la pente de la droite diminue, le point de contact [1] se décale vers la droite, et pour une certaine valeur de R_{TH} , il n'y a plus d'intersection entre les 2 courbes, et la température s'emballe.

Ce cas de figure sans point de contact est également vérifié, si pour V plus grand : la courbe $P = f(T)$ est située plus haut dans le quadrant, et se retrouve au-dessus de la droite $P = (T_J - T_A) / R_{TH}$.

De même, si la température ambiante augmente, la droite se translate vers la droite jusqu'à ne plus avoir de point commun, et l'emballement thermique est inévitable.


Pour finir, on comprend également l'importance de réduire R_{THJA} par l'ajout d'un dissipateur, qui permet une meilleure diffusion de la chaleur. En effet, si R_{TH} diminue, la pente de la droite augmente, ce qui rapproche la stabilité, et on peut même admettre T_A élevée.

Notons qu'il existe une technique connue : placer un ventilateur face au semi-conducteur de puissance, équipé généralement d'un dissipateur : on assure ainsi T_A voisin de la température de la pièce, soit une vingtaine de degrés par exemple.

3) Simulation Pspice de l'emballement thermique

3.a. Schéma

Remarque : sous Pspice, pour accéder, en cours de simulation, à un courant dans une branche, il est nécessaire de placer une source de tension nulle. Cette dernière a été ajoutée en série à Xdiode afin de disposer de l'information courant I_d , pour la source « H » (tension commandée par un courant) qui délivre $V(B)$.



On reprend le sous circuit Xdiode, alimenté sous 0,6 V.

Il nous faut ajouter la loi d'Ohm thermique. Pour ce faire, on capte le courant, la tension, et par un simple produit, on détermine la puissance. Il a été choisi de délivrer un courant, image de la puissance, puis, par loi d'Ohm avec une résistance électrique de valeur R_{TH} , d'avoir une tension image de ΔT .

La diode est placée à la température ambiante. On ajoute une tension dont la valeur en volt est T_A en degré pour aboutir à la température de jonction ($T_A + \Delta T$) qui entre dans le sous circuit.





schéma complet
netliste en [annexe3 emballement.cir](#)

Dans une simulation temporelle, il faut tenir compte de la capacité thermique. Dans l'analogie thermique / électrique, un condensateur est le composant qui « maintient » la température, comme en électricité, qui « maintient » la tension. En thermique, son unité est le $J/^\circ C$.

C_{TH} est placée en parallèle avec R_{TH} , et forme ainsi une constante de temps thermique de valeur $R_{TH} C_{TH}$. Si on prend arbitrairement 1 ms de constante de temps, cela donne, avec $R_{TH} = 73 \text{ }^\circ C/W$, $C_{TH} = 13,7 \text{ } \mu F$.

3.b. Stabilité

On vérifie la stabilité pour le point à $T_A = 60\text{ °C}$, $R_{TH} = 73\text{ °/W}$ (diode nue).



Tension, puissance et température de la diode

Interprétation

- A $t = 0$, pas de tension, donc pas de puissance, et $T_A = T_J = 60\text{ °C}$.


- On met sous tension $V(1) = 0,6\text{ V}$ à $t = 5\text{ ms}$. La puissance s'installe, la température de jonction croit. Après quelques ms, la température est proche de sa valeur finale.

A 30 ms, on est en régime établi : la température de jonction est de $63,165\text{ °C}$, la puissance est de $43,35\text{ mW}$. On retrouve les valeurs établies précédemment, au point [1] de fonctionnement stable.

X Values	2.0000m	30.000m
V(Temp)	60.000	63.165
W(Xdiode1)	0.000	43.351m
V(1)	0.000	600.000m

3.c. Instabilité par élévation de température ambiante

Supposons une température ambiante à 90 °C .




Tension, puissance et température de la diode

Interprétation

Avant $t = 5 \text{ ms}$, la diode est à $90 \text{ }^\circ\text{C}$, imposé par l'environnement.

A $t = 5 \text{ ms}$, un courant s'établit, ce qui fait dégager une puissance par la diode. Mais, avec le même R_{TH} , comme montré ci-contre, il n'y a plus de contact entre la courbe $P = f(T)$ et la droite $P = (T_J - T_A)/R_{TH}$.

Pas de position d'équilibre, et les courbes de puissance et température divergent.



$T_A = 90 \text{ }^\circ\text{C}$ et $R_{TH} = 73 \text{ }^\circ\text{C/W}$:
 La droite a conservé la même pente, mais s'est décalée vers la droite, perdant ainsi le point de contact : situation instable

Remarques :

- la température, issue de cette simulation, indique des valeurs de plusieurs centaines de $^\circ\text{C}$, ce qui dépasse les conditions de validité de la modélisation. En pratique, la diode est détruite...
- La durée de cette montée en température peut être différente en pratique : ici, c'est lié au choix de la constante de temps thermique.

3.d. Rôle bénéfique d'un dissipateur

Le boîtier TO220 est conçu pour y fixer un dissipateur.

Choisissons un modèle donné pour $R_{THSA} = 24,4 \text{ }^\circ\text{C/W}$, en air naturel.

On note :

A : ambiante, C : case, J : jonction, S : surface (ou Sink, pour dissipateur).

Le chemin entre la jonction et l'air ambiant montre 3 résistances en série, d'où :

$$R_{THJA} = R_{THJC} + R_{THCS} + R_{THSA}$$

Proposons $R_{THCS} = 1 \text{ }^\circ\text{C/W}$ (fonction du serrage, de l'état de surface).


Il vient $R_{THJA} = 3 + 1 + 24,4 = 28,4 \text{ }^\circ\text{C/W}$.



dissipateur
 19x13,2x12,7 mm
 à 24,4 $^\circ\text{C/W}$

La masse du dissipateur augmente la capacité thermique, ce qui ralentit le régime transitoire. Par souci de faire simple, nous allons proposer que le système reste du premier ordre, et que la constante de temps est passée arbitrairement à 3 ms. Soit $C_{TH} = 3 \text{ ms} / 28,4 = 106 \text{ } \mu\text{F}$. (Ces valeurs sont à placer dans la netlist pour une nouvelle simulation.). C'est la stabilisation à une valeur finale qui nous importe, et non le régime transitoire.

En relançant la simulation avec cette nouvelle valeur numérique, on obtient bien une montée en température plus lente et un fonctionnement de nouveau stable :




Tension, puissance et température de la diode à $T_A = 90 \text{ }^\circ\text{C}$ et $R_{TH} = 28,4 \text{ }^\circ\text{C/W}$.

Interprétation

A $t = 5 \text{ ms}$, un courant s'établit, ce qui fait dégager une puissance par la diode.

Avec $R_{TH} = 28,4 \text{ °C/W}$, et $T_A = 90 \text{ °C}$, nous avons un point de contact entre la courbe $P = f(T)$ et la droite $P = (T_J - T_A) / R_{TH}$, ce qui explique la stabilisation.

X Values	29.000m
V(Temp)	95.281
W(Xdiode1)	186.209m
V(1)	600.000m



$T_A = 90 \text{ °C}$ et $R_{TH} = 28,4 \text{ °C/W}$:
 La pente de la droite est plus élevée permettant ainsi le point de contact : situation stable


Remarque :

en toute rigueur, la température ambiante au voisinage du dissipateur est inférieure à précédemment, l'échange par convection ayant augmenté. Mais cela ne fait qu'améliorer la situation, la droite se décalant vers la gauche.

3.e. Rôle stabilisateur d'une résistance série

Une autre technique pour empêcher l'emballement thermique est de placer une résistance en série avec la diode. Bien entendu, sa valeur doit être la plus faible possible, pour éviter de pertes joules d'une part, et une chute de tension qui s'ajoute avec V_{AK} d'autre part.

Nous allons montrer, sur l'exemple précédent ($0,6 \text{ V}$ en direct, sans dissipateur, soit $R_{TH} = 73 \text{ °C/W}$, et $T_A = 90 \text{ °C}$), qu'une résistance de $5,5 \text{ m}\Omega$ est suffisante.




R_A , résistance additive, est placée en série avec la diode : elle est en série avec la résistance interne R_S .

On peut modifier la netliste pour placer la résistance additive. Mais, encore plus simple, on augmente la valeur de R_S , dont la valeur est $9,5 \text{ m}\Omega$.

Si R_S augmente, la courbe P « se couche », ce qui permet de raccrocher un point de fonctionnement.

Avec $R_S = 15 \text{ m}\Omega$, donc une résistance additive de $5,5 \text{ m}\Omega$, un point de contact existe, mais est tangent à la courbe $P = f(T)$ de la diode.



$T_A = 90 \text{ °C}$ et $R_{TH} = 73 \text{ °C/W}$:
 Selon la valeur de R_S , la courbe $P = f(T)$ permet un point de contact stable

La simulation à $R_S = 30 \text{ m}\Omega$, soit une résistance additive de $20,5 \text{ m}\Omega$ montre un meilleur cas.

CONCLUSION

Ce document a expliqué le phénomène d'emballement thermique sur une diode soumise à son auto-échauffement. Notons que ce problème est également connu sur certains montages à transistors bipolaires. Pour illustrer cela par la simulation, il a fallu adopter un modèle simplifié de diode, valable en polarisation directe. Les valeurs numériques sont donc cohérentes.

Un travail plus poussé demanderait un logiciel d'électrothermie, pour observer, au sein de la jonction PN, la diffusion de la chaleur en 3 D avec son environnement immédiat. Mais c'est un tout autre travail.

Annexe1 : fichier diode_temperature.cir

```

caracteristique directe
* fichier diode_temperature.cir
*.lib eval.lib

.model MUR810      D(Is=4.433p Rs=5.91m Ikf=92.2m N=1 Xti=3 Eg=1.11 Cjo=265.3p
+                M=.4253 Vj=.75 Fc=.5 Isr=49.76n Nr=2 Tt=61.66n T_ABS={Tdiode})
*                Motorola   pid=MUR805 case=TO220AC

* pour MUR810 simplifiée
.param IS=5.5p
.param N=1.095
.param EG=1.11
.param XTI=4
.param RS=9.5m
.param Tdiode=27

.MODEL Dtemoin D (IS={IS} N={N} EG={EG} XTI={XTI} RS={RS} T_ABS={Tdiode} )

* pour sous-circuit DiodeT, commandée en température
.param Tnom=300 ; en K pour les équations
.param k=1.38E-23
.param q=1.6E-19
.param kq={k}/{q}
.param XTIN={{XTI}/{N}}

.SUBCKT DiodeT A K T+ T- ; a la ddp T+ - T-, les Volt représentent les Celsius
ETK TK 0 POLY(1) (T+,T-) 273 1 ; en TK, les Volt représentent les Kelvin
EVT VT 0 POLY(1) (TK,0) 0 {kq}; tension VT=kT/q en S.I. 26 mV à 27°C
Rserie A Ai {RS}
Gexp Ai K VALUE={ IS*(exp((V(Ai)-V(K))/(N*V(VT))))-1
+ * ( exp (
+ ((V(TK)/Tnom) -1)
+ * ( 1.11 - 702E-6 * PWR(V(TK),2) / (V(TK)+1108) )
+ / (N*V(VT) ) )
+ * PWR( (V(TK)/Tnom),XTIN ) )
+ }
.ENDS

* Circuit

Vtemp com+ com- DC={Tdiode} ; pour injecter une température en degré Celsius
Rbidon+ com+ 0 1Meg
Rbidon- com- 0 1Meg

Vdiode 1 0
Dmur 1 0 MUR810 ; vrai diode
Xdiode 1 0 com+ com- DiodeT ; diode maison
D1 1 0 Dtemoin ; identique diode maison FACULTATIF

.OP
.DC Vdiode 0.2 1.1 1m
.STEP param Tdiode list 25 100 175

.probe
.end

```

Netliste, prête à simuler

Annexe2 : fichier diode_commandee.cir

```

diode avec une attaque en temperature
* fichier diode_commandee.cir

.model MUR810      D(Is=4.433p Rs=5.91m Ikf=92.2m N=1 Xti=3 Eg=1.11 Cjo=265.3p
+                M=.4253 Vj=.75 Fc=.5 Isr=49.76n Nr=2 Tt=61.66n)
*                Motorola      pid=MUR805 case=TO220AC

* pour MUR810 simplifiée
.param IS=5.5p
.param N=1.095
.param EG=1.11
.param XTI=4
.param RS=9.5m

* pour sous circuit Xdiode
.param Tnom=300 ; en K pour les équations
.param k=1.38E-23
.param q=1.6E-19
.param kq={{k}/{q}}
.param XTIN={{XTI}/{N}}

.SUBCKT DiodeT A K T+ T- ; a la ddp T+ - T-, les Volt représentent les Celsius
ETK TK 0 POLY(1) (T+,T-) 273 1 ; en TK, les Volt représentent les Kelvin
EVT VT 0 POLY(1) (TK,0) 0 {kq}; tension VT=kT/q en S.I. 26 mV à 27°C
Rserie A Ai {RS}
Gexp Ai K VALUE={ {IS}*(exp((V(Ai)-V(K))/({N}*V(VT))))-1)
+ * ( exp (
+ ((V(TK)/{Tnom}) -1)
+ * ( 1.11 - 702E-6 * PWR(V(TK),2) / (V(TK)+1108) )
+ / ({N}*V(VT) ) )
+ * PWR( (V(TK)/{Tnom}),XTIN ) )
+ }
.ENDS

Vtemp com+ com- EXP (25 175 0.1 0.1 1 0.1) DC=0 ; pour injecter en degré Celsius
Rbidon+ com+ 0 1Meg
Rbidon- com- 0 1Meg

Vdiode 1 0 DC=0.6
Xdiode 1 0 com+ com- DiodeT
D1 1 0 MUR810

.OP
.DC Vtemp 0 175 0.1
*.TRAN 0.01 1.5 0 0.01
.probe
.end

```

Netliste, prête à simuler

Annexe3 : fichier `emballement.cir.cir`

```

emballement thermique
* fichier emballement.cir
.lib eval.lib

.param IS=5.5p
.param N=1.095
.param EG=1.11
.param XTI=4
.param RS=9.5m

* diode commandée par température
.param Tnom=300 ; en K pour les équations
.param k=1.38E-23
.param q=1.6E-19
.param kq={k}/{q}
.param XTIN={XTI}/{N}

.SUBCKT DiodeT A K T+ T- ; a la ddp T+ - T-, les Volt représentent les Celsius
ETK TK 0 POLY(1) (T+,T-) 273 1 ; en TK, les Volt représentent les Kelvin
EVT VT 0 POLY(1) (TK,0) 0 {kq}; tension VT=kT/q en S.I. 26 mV à 27°C
Rserie A Ai {RS}
Gexp Ai K VALUE={ IS*(exp((V(Ai)-V(K))/(N*V(VT)))-1)
+ * ( exp (
+ ((V(TK)/Tnom) -1)
+ * ( 1.11 - 702E-6 * PWR(V(TK),2) / (V(TK)+1108) )
+ / (N*V(VT) ) )
+ * PWR( (V(TK)/Tnom),XTIN ) )
+ }
.ENDS

Vdiode 1 0 DC=1 PWL (0,0 5m,0 5.01m 0.6)

Xdiode1 1 Ki Temp 0 DiodeT ; com+ com- DiodeT
Vnul Ki 0 DC=0 ; parcouru par Idiode

HU1 B 0 POLY(1) Vnul 0 1 ; délivre tension image de Id
Gmult 1 4 POLY(2) (1,0) (B,0) 0 0 0 0 1 ; délivre un courant image VAK Id
Rf 4 0 {RTH} ;
CTH 4 0 13.7u ; 1 ms de constante de temps

.param RTH=73
.param TA=60
VAMB Temp 4 DC={TA} ; forme un décalage

.IC V(4)=0
.OP
.TRAN 10u 40m 0 10u
.probe
.end

```

Netliste, prête à simuler

articles 1 à 43 : sur le livre

Tableau récapitulatif des articles PDF disponibles sur ce site

n°	titre	lien présentation	lien direct article
	Guide d'installation et d'emploi simplifié	présentation	document PDF
44	Exemples basiques et des exercices...	présentation	document PDF
45	Un exemple de circuit passif	présentation	document PDF
46	Un oscillateur Colpitts	présentation	document PDF
47	Compensation en fréquence des amplificateurs opérationnels	présentation	document PDF
48	Un amplificateur à transistors bipolaires	présentation	document PDF
49	Une bascule D Flip Flop CMOS	présentation	document PDF
50	Une porte XOR à transistors MOS	présentation	document PDF
51	Un VCO à 12 transistors MOS	présentation	document PDF
52	Une PLL à moins de 20 transistors MOS	présentation	document PDF
53	Un oscillateur à résistance négative	présentation	document PDF
54	Une charge électronique	présentation	document PDF
55	Un amplificateur en classe C	présentation	document PDF
56	Le monostable 74 123	présentation	document PDF
57	Un amplificateur en classe D	présentation	document PDF
58	Le transformateur en linéaire	présentation	document PDF
59	La loi d'ohm thermique	présentation	document PDF
60	Le transformateur en non linéaire	présentation	document PDF
61	Robustesse d'un oscillateur en anneau	présentation	document PDF
62	Une alimentation stabilisée	présentation	document PDF
63	Modélisation d'un haut-parleur	présentation	document PDF
64	Un synthétiseur de fréquence	présentation	document PDF
65	Un ampli audio de Sparkfun	présentation	document PDF
66	Simulation logique et analogique	présentation	document PDF
67	Un oscillateur à relaxation	présentation	document PDF
68	Lecteur de TAG RFID 125 kHz	présentation	document PDF
69	Diagramme de l'œil avec Pspice	présentation	document PDF
70	Un amplificateur hautes fréquences	présentation	document PDF
71	Une bizarrerie enfin expliquée...	présentation	document PDF
72	Comprendre le paramétrage de la FFT	présentation	document PDF
73	La relation de Bennett	présentation	document PDF
74	Simuler un circuit à plus de 20 transistors avec PSpice Eval	présentation	document PDF
75	Une horloge biphase sans recouvrement	présentation	document PDF
76	Quelques simulations sur la diode	présentation	document PDF
77	Un ampli classe A, avec transformateur de sortie	présentation	document PDF
78	Des stimuli pour PSpice	présentation	document PDF
79	Simuler le TL431 : zener ajustable	présentation	document PDF
80	Un ADC flash	présentation	document PDF
81	Une chaîne d'acquisition : S&H, ADC, DAC	présentation	document PDF
82	Un amplificateur 50 MHz	présentation	document PDF
83	Un dérivateur non inverseur	présentation	document PDF
84	Un amplificateur bipolaire avec push pull CMOS	présentation	document PDF
85	Rôle des répéteurs logiques dans un circuit intégré	présentation	document PDF
86	Un driver logique CMOS pour charge 50 ohms	présentation	document PDF
87	Des triggers de Schmitt et des applications	présentation	document PDF
88	Un filtre gaussien analogique	présentation	document PDF
89	Un générateur de bruit rose	présentation	document PDF

90	Un anémomètre à fil chaud : simulation comportementale	présentation	document PDF
91	Un oscillateur à pont de Wien stabilisé par CTN	présentation	document PDF
92	L'emballement thermique d'une diode	présentation	document PDF
93	Les puissances dans un amplificateur	présentation	document PDF
94	Asservissement de puissance dans une résistance	présentation	document PDF
95	Asservissement de la puissance émise par une antenne radio	présentation	document PDF
96	Un driver de LED de puissance	présentation	document PDF
97	Exploiter Pspice pour simuler des filtres numériques	présentation	document PDF
98	Un filtre en cosinus surélevé avec Pspice	présentation	document PDF
99	Effet de la température sur un amplificateur en classe A	présentation	document PDF
100	Un amplificateur à transistors JFET et bipolaires	présentation	document PDF
Supplément, hors article :			
mon cours « Electronique pour les communications numériques », polycopié couleur 201 pages en pdf			

[retour à l'écran d'accueil de ce site](#)